Math 132
Fall 2006 Exam II

1. Suppose that $f(x) = 2^{(x^2)}$. Calculate D(f)(2).

a) 8 ln(2) b) 16 ln(2) c) 32 ln(2) d) 64 ln(2) e) 128 ln(2)
f) 8/ln(2) g) 16/ln(2) h) 32/ln(2) i) 64/ln(2) j) 128/ln(2)

Solution: d

```
> A := diff(2^(x^2), x);
```

$$A := 2\,2^{(x^2)}\,x\ln(2)$$

```
> subs(x=2, A);
```

$$64\ln(2)$$

2. Calculate $\displaystyle\int_1^2 \log_2(x)\,dx$.

a) $2 - \dfrac{1}{\ln(2)}$ b) $2 + \dfrac{1}{\ln(2)}$ c) $2 - \ln(2)$ d) $2 + \ln(2)$ e) $2\ln(2) - 1$

f) $2\ln(2) + 1$ g) $1 - \dfrac{2}{\ln(2)}$ h) $1 + \dfrac{2}{\ln(2)}$ i) $\log_2(2)$ j) $\log_2(2) - 1$

Solution: a

```
> int(ln(x)/ln(2), x = 1 .. 2);   #log[2](x) = ln(x)
```

$$\frac{-1 + 2\ln(2)}{\ln(2)}$$

```
> testeq(int(log[2](x),x=1..2) = 2 - 1/ln(2)); #Check answer.
```

true

3. Suppose that $f(x) = \sqrt{x}^{\,x}$. Calculate D(f)(2).

a) $\sqrt{2}+1$ b) $2\sqrt{2}+1$ c) $4\sqrt{2}+1$ d) $\sqrt{2}+\ln(2)$ e) $2\sqrt{2}+\ln(2)$
f) $4\sqrt{2}+\ln(2)$ g) $\ln(2)+1$ h) $\ln(2)+2$ i) $2\ln(2)+1$ j) $\sqrt{2}\ln(2)$

Solution: g

```
> diff(sqrt(x)^x, x);
```

$$(\sqrt{x})^x\left(\ln(\sqrt{x})+\frac{1}{2}\right)$$

```
> simplify(subs(x=2, %));
```

$$\ln(2)+1$$

```
> eqn1 := f(x)  = sqrt(x)^x;
  #The next steps verify by logarithmic differentiation
```

$$eqn1 := f(x)=(\sqrt{x})^x$$

```
> eqn2 := map(ln, eqn1);
```

$$eqn2 := \ln(f(x))=\ln((\sqrt{x})^x)$$

```
> eqn3 := map( z -> diff(z,x), eqn2);
```

$$eqn3 := \frac{\frac{d}{dx}f(x)}{f(x)}=\ln(\sqrt{x})+\frac{1}{2}$$

```
> eqn4 := diff(f(x),x) = solve(eqn3, diff(f(x),x));
```

$$eqn4 := \frac{d}{dx}f(x)=\ln(\sqrt{x})\,f(x)+\frac{1}{2}f(x)$$

```
> eqn5 := subs(eqn1, eqn4);
```

$$eqn5 := \frac{d}{dx}(\sqrt{x})^x=\ln(\sqrt{x})\,(\sqrt{x})^x+\frac{(\sqrt{x})^x}{2}$$

```
> answer := subs(x = 2, rhs(eqn5));
```

$$answer := 2\ln(\sqrt{2})+1$$

```
> simplify(answer);
```

$$\ln(2)+1$$

4. A radioactive substance has mass 120g at time $t=8$ and mass 80g at time $t=10$. Express the half-life of this substance in terms of the quantities $\alpha=\ln(2)$ and $\beta=\ln(3)$.

a) $\beta-\alpha$ b) $2\beta-\alpha$ c) $\frac{\alpha}{\beta-\alpha}$ d) $\frac{\beta}{\beta-\alpha}$ e) $\frac{2\alpha}{\beta-\alpha}$

f) $\frac{2\beta}{\beta-\alpha}$ g) $\frac{\beta+\alpha}{\beta-\alpha}$ h) $\frac{\beta-\alpha}{\beta}$ i) $\frac{\beta-\alpha}{\alpha}$ j) $\frac{2\beta+\alpha}{2\beta-\alpha}$

Solution: e

```
> m := t -> m0/2^(t/tau);
```

$$m := t \to \frac{m0}{2^{\left(\frac{t}{\tau}\right)}}$$

```
> eqn1 := m(8) = 120;
```

$$eqn1 := \frac{m0}{2^{\left(\frac{8}{\tau}\right)}} = 120$$

```
> eqn2 := m(10) = 80;
```

$$eqn2 := \frac{m0}{2^{\left(\frac{10}{\tau}\right)}} = 80$$

```
> solve({eqn1,eqn2},{m0,tau});
```

$$\left\{\tau = -\frac{2\ln(2)}{\ln\left(\frac{2}{3}\right)}, m0 = \frac{1215}{2}\right\}$$

```
> eqn3 := m0 = solve(eqn1, m0); #Step=by-step calculation of tau
```

$$eqn3 := m0 = 120\, 2^{\left(\frac{8}{\tau}\right)}$$

```
> eqn4 := subs(eqn3, eqn2);
```

$$eqn4 := \frac{120\, 2^{\left(\frac{8}{\tau}\right)}}{2^{\left(\frac{10}{\tau}\right)}} = 80$$

```
> eqn5 := map(z-> z*2^(10/tau)/2^(8/tau), eqn4);
```

$$eqn5 := 120 = \frac{80\, 2^{\left(\frac{10}{\tau}\right)}}{2^{\left(\frac{8}{\tau}\right)}}$$

```
> eqn6 := map(simplify, eqn5);
```

$$eqn6 := 120 = 80\, 4^{\left(\frac{1}{\tau}\right)}$$

```
> eqn7 := map(ln, eqn6);
```

$$eqn7 := \ln(120) = \ln\left(80\, 4^{\left(\frac{1}{\tau}\right)}\right)$$

```
> eqn8 := expand(eqn7) assuming tau > 0;
```

$$eqn8 := \ln(120) = \ln(80) + \frac{\ln(4)}{\tau}$$

```
> eqn9 := tau = solve(eqn8, tau);
```

$$eqn9 := \tau = \frac{\ln(4)}{\ln(120) - \ln(80)}$$

```
> eqn10 := simplify(eqn9);
```

$$eqn10 := \tau = -\frac{2\ln(2)}{\ln(2) - \ln(3)}$$

5. The mass m(t) of a microbe colony feasting in a gourmet nutrient broth doubles every 2 hours. When the colony's mass is 6 µg, what, in µg/hr, is D(m)(t) ?

a) ln(2)   b) 2 ln(2)   c) 3 ln(2)   d) 6 ln(2)   e) 12 ln(2)
f) 1/ln(2)   g) 2/ln(2)   h) 3/ln(2)   i) 6/ln(2)   j) 12/ln(2)

Solution: c

```
> subs(t=0, diff(6*2^(t/2), t));
```

$$3\ln(2)$$

```
> m := t -> m0*2^(t/2);
```

$$m := t \rightarrow m0\, 2^{(1/2\, t)}$$

```
> D(m)(t);
```

$$\frac{1}{2}\, m0\, 2^{\left(\frac{t}{2}\right)} \ln(2)$$

```
> 'D(m)(t)' = 'm(t)'*ln(2)/2;
#Notice that m(t) = m0*2^(t/2) is a factor of D(m)(t)
```

$$D(m)(t) = \frac{1}{2}\, m(t) \ln(2)$$

```
> answer := (1/2)*6*ln(2);
```

$$answer := 3\ln(2)$$

6. If $u(t)$ is the unique solution of the initial value problem $\frac{d}{dt}u(t) = 20 - 5\,u(t),\ u(0) = 3,$

then what is $\lim_{t \to \infty} u(t)$?

a) 1 b) 2 c) 3 d) 4 e) 5
f) 6 g) 10 h) 12 i) 15 j) 20

Solution: d

```
> dsolve({diff(u(t),t) = 20-5*u(t), u(0) = 3},u(t));
```

$$u(t) = 4 - e^{(-5\,t)}$$

```
> limit(rhs(%), t = infinity);
```

$$4$$

7. Suppose that $f(x) = \arcsin(x)$. Calculate $D(f)\left(\frac{\sqrt{3}}{2}\right)$. (The derivative of f(x) at $x = \frac{\sqrt{3}}{2}$).

a) 1 b) $\sqrt{2}$ c) $\frac{1}{\sqrt{2}}$ d) $\sqrt{3}$ e) $\frac{2}{\sqrt{3}}$
f) $\frac{1}{2}$ g) 2 h) $2\sqrt{2}$ i) $2\sqrt{3}$ j) 4

Solution: g

```
> D(arcsin)(sqrt(3)/2);
```

$$\sqrt{4}$$

8. Suppose that $f(x) = -\arctan\left(\frac{1}{x}\right)$. What is $D(f)(\sqrt{3})$? (The derivative of f(x) at $x = \sqrt{3}$).

a) $\frac{1}{2}$ b) $\frac{1}{3}$ c) $\frac{2}{3}$ d) $\frac{1}{4}$ e) $\frac{1}{\sqrt{3}}$

f) $\frac{\sqrt{3}}{2}$ g) $\sqrt{3}$ h) $2\sqrt{3}$ i) 2 j) 4

Solution: d

```
> D(x -> -arctan(1/x))(sqrt(3));
```

$$\frac{1}{4}$$

9. Calculate $\int_0^{\frac{\pi}{2}} x \sin(x)\, dx$.

a) $\frac{1}{4}$ b) $\frac{1}{2}$ c) $\frac{3}{4}$ d) 1 e) 2

f) $\frac{\pi}{4}$ g) $\frac{\pi}{2}$ h) $\frac{3\pi}{4}$ i) π j) 2π

Solution: d

```
> int(x*sin(x),x=0..Pi/2);
```

$$1$$

10. Calculate $\int_0^1 x^2 e^{(-x)}\, dx$.

a) $1-\frac{1}{e}$ b) $1-\frac{2}{e}$ c) $1-\frac{3}{e}$ d) $1-\frac{4}{e}$ e) $1-\frac{5}{e}$

f) $2-\frac{1}{e}$ g) $2-\frac{2}{e}$ h) $2-\frac{3}{e}$ i) $2-\frac{4}{e}$ j) $2-\frac{5}{e}$

Solution: j

```
> int(x^2*exp(-x),x = 0 .. 1);
```

$$-5\,e^{(-1)}+2$$

11. Calculate $16\int_1^e x^3 \ln(x)\,dx$.

a) e^4-1 b) $2e^4-1$ c) $3e^4-1$ d) $4e^4-1$ e) $5e^4-1$

f) e^4+1 g) $2e^4+1$ h) $3e^4+1$ i) $4e^4+1$ j) $5e^4+1$

Solution: h

```
> 16*int(x^3*ln(x),x = 1 .. exp(1));
```

$$3\,e^4+1$$

12. Calculate $\int_1^2 \frac{3x+2}{x^2+x}\,dx$.

a) $\ln(2)$ b) $\ln(3)$ c) $2\ln(2)$ d) $2\ln(3)$ e) $3\ln(2)$

f) $\ln(6)$ g) $\ln\left(\frac{2}{3}\right)$ h) $\ln\left(\frac{3}{2}\right)$ i) $\ln\left(\frac{9}{2}\right)$ j) $\ln(12)$

Solution: f

```
> normal(1/x+2/(x+1) - 1/(x+1)^2);
```

$$\frac{3x^2+3x+1}{x\,(x+1)^2}$$

```
> testeq(int((3*x+2)/(x^2+x), x= 1..2) = ln(6));
```

true

13. Calculate $\int_1^2 \frac{3x^2+3x+1}{x\,(x+1)^2}\,dx$.

a) $\ln(2)-\frac{1}{3}$ b) $\ln(3)-\frac{1}{2}$ c) $2\ln(2)-\frac{1}{6}$ d) $2\ln(3)-\frac{1}{4}$ e) $3\ln(2)-\frac{1}{2}$

f) $\ln(6)-\frac{2}{3}$ g) $\ln\left(\frac{3}{2}\right)-\frac{1}{4}$ h) $\ln\left(\frac{3}{2}\right)-\frac{1}{3}$ i) $\ln\left(\frac{9}{2}\right)-\frac{1}{6}$ j) $\ln\left(\frac{9}{4}\right)-\frac{1}{3}$

Solution: i

```
> A := int((3*x^2+3*x+1)/x/((x+1)^2),x = 1 .. 2);
```

$$A := -\frac{1}{6}+2\ln(3)-\ln(2)$$

```
> testeq(A = ln(9/2)-1/6);
```

true

14. Calculate $\int_0^{\frac{\pi}{2}} \sin(x)^2 \cos(x)^3 \, dx.$

a) 1/15 b) 2/15 c) 1/5 d) 4/15 e) 1/3
f) 2/5 g) 7/15 h) 8/15 i) 3/5 j) 2/3

Solution: b

```
> int(sin(x)^2*cos(x)^3,x = 0 .. Pi/2);
```

$$\frac{2}{15}$$

15. From the reduction formula

$$\int \cos(x)^n \, dx = \frac{\sin(x)\cos(x)^{(n-1)}}{n} + \frac{(n-1)\int \cos(x)^{(n-2)} \, dx}{n}$$

it follows that there are rational numbers A, B, and C for which

$$\int \cos(x)^8\, dx = A \sin(x)\cos(x)^7 + B \sin(x)\cos(x)^5 + C \int \cos(x)^4\, dx$$

What is C?

a) 8/15	b) 2/3	c) 7/9	d) 35/48	e) 5/7
f) 24/35	g) 16/21	h) 9/16	i) 3/4	j) 3/5

Solution: d

```
> eqn1 := J(n)= 1/n*sin(x)*cos(x)^(n-1) + (n-1)/n*J(n-2);
```

$$eqn1 := \mathrm{J}(n) = \frac{\sin(x)\cos(x)^{(n-1)}}{n} + \frac{(n-1)\,\mathrm{J}(n-2)}{n}$$

```
> eqn2 := subs(n=n-2, eqn1);
```

$$eqn2 := \mathrm{J}(n-2) = \frac{\sin(x)\cos(x)^{(n-3)}}{n-2} + \frac{(n-3)\,\mathrm{J}(n-4)}{n-2}$$

```
> eqn3 := subs(eqn2, eqn1);
```

$$eqn3 := \mathrm{J}(n) = \frac{\sin(x)\cos(x)^{(n-1)}}{n} + \frac{(n-1)\left(\frac{\sin(x)\cos(x)^{(n-3)}}{n-2} + \frac{(n-3)\,\mathrm{J}(n-4)}{n-2}\right)}{n}$$

```
> eqn4 := subs(n=8, eqn3);
```

$$eqn4 := \mathrm{J}(8) = \frac{1}{8}\sin(x)\cos(x)^7 + \frac{7}{48}\sin(x)\cos(x)^5 + \frac{35}{48}\mathrm{J}(4)$$

```
> testeq(cos(x)^8 = diff(
1/8*sin(x)*cos(x)^7+7/48*sin(x)*cos(x)^5+35/48*int(cos(x)^4,x),
x));
```

$$true$$

16. Calculate $\displaystyle\int_0^{\frac{1}{\sqrt{2}}} \frac{x^2}{\sqrt{1-x^2}}\, dx.$

a) $\frac{\pi-2}{8}$ b) $\frac{\pi-2}{4}$ c) $\frac{\pi-2}{2}$ d) $\frac{\pi-1}{8}$ e) $\frac{\pi-1}{4}$

f) $\frac{\pi-1}{2}$ g) $\frac{4-\pi}{8}$ h) $\frac{4-\pi}{4}$ i) $\frac{4-\pi}{2}$ j) $\frac{8-\pi}{8}$

Solution: a

```
> A := normal(2/x-3/(x^2+1));
```

$$A := \frac{2x^2+2-3x}{x(x^2+1)}$$

```
> int(x^2/sqrt(1-x^2),x=0..1/sqrt(2));
```

$$-\frac{1}{4}+\frac{\pi}{8}$$

17. Calculate $\int_0^1 \frac{1}{\sqrt{1+x^2}}\,dx$.

a) $\frac{1}{2}\ln(2)$ b) $\ln(2)$ c) $2\ln(2)$ d) $\frac{1}{2}\sqrt{1+\ln(2)}$ e) $\sqrt{1+\ln(2)}$

f) $2\ln(3)$ g) $\frac{1}{2}\ln\left(1+\frac{1}{\sqrt{2}}\right)$ h) $\ln\left(1+\frac{1}{\sqrt{2}}\right)$ i) $\frac{1}{2}\ln(1+\sqrt{2})$ j) $\ln(1+\sqrt{2})$

Solution: j

```
> testeq(int(1/sqrt(1+x^2),x = 0 .. 1)=ln(sqrt(2)+1));
```

true

```
>
>
```

18. There are unique rational numbers A, B, and C such that

$$\int_1^2 \frac{2x^2+2-3x}{x(x^2+1)}\,dx = A\arctan(2)+B\ln(2)+C\pi.$$ **What is A ?**

a) –5 b) –4 c) –3 d) –2 e) –1

f) 5 g) 4 h) 3 i) 2 j) 1

Solution: c

```
> int((2*x^2+2-3*x)/x/(x^2+1),x=1..2);
```

$$-3\arctan(2) + 2\ln(2) + \frac{3\pi}{4}$$

```
> convert((2*x^2+2-3*x)/x/(x^2+1), parfrac, x);
```

$$-\frac{3}{x^2+1} + \frac{2}{x}$$

19. The region in the first quadrant bounded above by $y = 3 - x^2$ and below by $y = 2x$ is rotated about the x-axis. What is the volume of the resulting solid of revolution?

a) $\frac{88\pi}{15}$ b) $\frac{92\pi}{15}$ c) $\frac{33\pi}{5}$ d) $\frac{20\pi}{3}$ e) $\frac{36\pi}{5}$

f) $\frac{112\pi}{15}$ g) $\frac{40\pi}{3}$ h) $\frac{48\pi}{5}$ i) $\frac{148\pi}{15}$ j) 10π

Solution: a

```
> Pi*int((3-x^2)^2-(2*x)^2,x=0..1);
```

$$\frac{88\pi}{15}$$

```
>
```

20. The region above the x-axis and under the graph of $y = \frac{\sin(x)}{x}$, $0 < x \le \frac{\pi}{2}$ is rotated about the y-axis. What is the volume of the resulting solid of revolution?

a) $\frac{\pi}{6}$ b) $\frac{\pi}{4}$ c) $\frac{\pi}{3}$ d) $\frac{\pi}{2}$ e) $\frac{3\pi}{4}$

f) π g) $\frac{3\pi}{2}$ h) 2π i) 3π j) 4π

Solution: h

```
> int(2*Pi*x*sin(x)/x,x=0..Pi/2);
```

$$2\,\pi$$

```
>
```